



UNIÓN INTERNACIONAL DE TELECOMUNICACIONES

CCITT

COMITÉ CONSULTIVO
INTERNACIONAL
TELEGRÁFICO Y TELEFÓNICO

Z.100 Anexo C1

(11/1988)

SERIE Z: LENGUAJES Y ASPECTOS GENERALES DE
SOPORTE LÓGICO PARA SISTEMAS DE
TELECOMUNICACIÓN

Lenguaje de especificación y descripción funcionales
(LED)

Criterios para la utilización de técnicas de descripción
formal (TDF)

RESUMEN DE LA SINTAXIS LED/GR

Reedición de la Recomendación Z.100 Anexo C1 del
CCITT publicada en el Libro Azul, Fascículo X.1 (1988)

NOTAS

- 1 La Recomendación Z.100 Anexo C1 del CCITT se publicó en el fascículo X.1 del Libro Azul. Este fichero es un extracto del Libro Azul. Aunque la presentación y disposición del texto son ligeramente diferentes de la versión del Libro Azul, el contenido del fichero es idéntico a la citada versión y los derechos de autor siguen siendo los mismos (véase a continuación).
- 2 Por razones de concisión, el término «Administración» se utiliza en la presente Recomendación para designar a una administración de telecomunicaciones y a una empresa de explotación reconocida.

Resumen de la sintaxis LED/GR

C1.1 *Introducción*C1.1.1 *Metalinguaje*

Para la gramática gráfica, el metalenguaje descrito en la Recomendación LED (§ 1.5.2) se amplía con los siguientes metasímbolos:

- a) **contains**
- b) **is associated with**
- c) **is followed by**
- d) **is connected to**
- e) **set**

El metasímbolo **set** es un operador postfijo que actúa sobre los elementos sintácticos que le preceden inmediatamente dentro de llaves, e indica un conjunto (no ordenado) de ítems. Cada uno de estos ítems puede ser cualquier elemento sintáctico, en cuyo caso deberá aplicarse antes del metasímbolo **set**. Ejemplo:

{ { <área de texto de sistema> } * { <diagrama de macro> } * <área de interacción de bloques> } set


es un conjunto de cero o más <área de texto de sistemas>s, cero o más <diagrama de macro>s y un <área de interacción de bloques>.

Todos los demás metasímbolos son operadores infijos que tienen un símbolo gráfico no terminal como argumento de la izquierda. El argumento de la derecha es, bien un grupo de elementos sintácticos dentro de llaves, bien un solo elemento sintáctico. Si el lado derecho de una regla de producción tiene un símbolo gráfico no terminal como primer elemento y contiene uno o más de estos operadores infijos, el símbolo gráfico no terminal es el argumento de la izquierda de cada uno de los operadores infijos mencionados. Un símbolo gráfico no terminal es un no terminal que tiene la palabra «símbolo» inmediatamente después del signo >.


El metasímbolo **contains** indica que su argumento de la derecha debe colocarse dentro de su argumento de la izquierda y del <símbolo de ampliación de texto> asociado, si existe. Ejemplo:

<referencia de bloque> ::=
 <símbolo de bloque> **contains** <nombre de bloque>

<símbolo de bloque> ::



significa lo siguiente:



El metasímbolo **is associated with** indica que su argumento de la derecha está asociado lógicamente con su argumento de la izquierda (como si estuviese «contenido» en ese argumento; la asociación inequívoca se asegura por reglas de dibujo apropiadas).

El metasímbolo **is followed by** significa que su argumento de la derecha sigue (tanto lógicamente como en el dibujo) a su argumento de la izquierda.

El metasímbolo **is connected to** significa que su argumento de la derecha está conectado (tanto lógicamente como en el dibujo) a su argumento de la izquierda.

C1.1.2 Reglas generales

C1.1.2.1 Partición de diagramas

La siguiente definición de partición de diagramas no forma parte de la *gramática gráfica concreta*, pero se utiliza el mismo metalenguaje.

<página> ::=
 <símbolo de casilla> **contains**
 <área de encabezamiento> <área de número de página>
 { <unidad sintáctica> }*

<área de encabezamiento> ::=
 <símbolo de texto implícito> **contains** <encabezamiento>

<área de número de página> ::=
 <símbolo de texto implícito> **contains** [<número de página> [(<número de página>)]]

<número de página> ::=
 <nombre de literal>

<número de páginas> ::=
 <nombre de literal natural>


<página> es un no terminal comienzo, por lo que no está referido a ninguna regla de producción. Un diagrama puede particionarse en varias <páginas>s, en cuyo caso el <símbolo de casilla> que delimita el diagrama y el <encabezamiento> del diagrama se reemplazarán por un <símbolo de casilla> y un <encabezamiento> para cada página.

El <símbolo de texto implícito> no se muestra pero está implícito para obtener una clara separación entre <área de encabezamiento> y <área de número de página>. <área de encabezamiento> se sitúa en la esquina superior izquierda del <símbolo de casilla>. <área de número de página> se sitúa en la esquina superior derecha del <símbolo de casilla>. <encabezamiento> y <unidad sintáctica> dependen del tipo de diagrama.

C1.1.2.2 Comentario

<área de comentario> ::=
 <símbolo de comentario> **contains** <texto>
 is connected to <símbolo de asociación de trazo discontinuo>

<símbolo de comentario> ::=



<símbolo de asociación de trazo discontinuo> ::=

Un extremo del <símbolo de asociación de trazo discontinuo> debe conectarse al centro del segmento vertical del <símbolo de comentario>.

Un <símbolo de comentario> se conecta a cualquier símbolo gráfico por medio de un <símbolo de asociación de trazo discontinuo>. El <símbolo de comentario> se considera un símbolo cerrado completando (en la imaginación) el rectángulo. Contiene texto de comentario relacionado con el símbolo gráfico.

C1.1.2.3 Ampliación de texto

<área de ampliación de texto> ::=
 <símbolo de ampliación de texto> **contains** <texto>
 is connectd to <símbolo de asociación de trazo continuo>

<símbolo de ampliación de texto> ::
 <símbolo de comentario>

<símbolo de asociación de trazo continuo> ::=

Un <símbolo de ampliación de texto> se conecta a cualquier símbolo gráfico por medio de un <símbolo de asociación de trazo continuo>. El <símbolo de ampliación de texto> se considera un símbolo cerrado completando (en la imaginación) el rectángulo.

Un extremo del <símbolo de asociación de trazo continuo> debe estar conectado al centro del segmento vertical del <símbolo de ampliación de texto>.

El texto contenido en el <signo de ampliación de texto> es continuación del texto dentro del símbolo gráfico y se considera contenido en ese símbolo.

C1.2 Definición de sistema

<definición de sistema concreto> ::=
 { <definición de sistema> | <diagrama de sistema> } { <definición remota> }*

<definición remota> ::=
 <definición>
 | <diagrama>

<diagrama> ::=
 <diagrama de bloque>
 | <diagrama de proceso>
 | <diagrama de procedimiento>
 | <diagrama de subestructura de bloque>
 | <diagrama de subestructura de canal>
 | <diagrama de servicio>
 | <diagrama de macro>

C1.3 Diagrama de sistema

<diagrama de sistema> ::=
 <símbolo de casilla> **contains**
 { <encabezamiento de sistema>
 { <área de texto de sistema> }*
 { <diagrama de macro> }*
 <área de interacción de bloques> }set }

<símbolo de casilla> ::=



<encabezamiento de sistema> ::=
 SYSTEM <nombre de sistema>

<área de texto de sistema> ::=
 <símbolo de texto> **contains**
 { <definición de señal>
 | <definición de lista de señales>
 | <definición de datos>
 | <definición de macro>
 | <definición de seleccionar> }*

<símbolo de texto> ::=



```

<área de interacción de bloques> ::=
    { <área de bloque> | <área de definición de canal> }+

<área de bloque> ::=
    <referencia gráfica de bloque>
    | <diagrama de bloque>

<referencia gráfica de bloque> ::=
    <símbolo de bloque> contains <nombre de bloque>

<símbolo de bloque> ::=

```



```

<área de definición de canal> ::=
    <símbolo de canal>
    is associated with { <nombre de canal>
        { [ { <identificador de canal> | <identificador de bloque> } ]
          <área de lista de señales>
          [ <área de lista de señales> ] } set }
    is connected with { <área de bloque>
        { <área de bloque> | <símbolo de casilla> }
        [ <área de asociación de subestructura de canal> ] } set

```

El <identificador de canal> identifica un canal externo conectado al <diagrama de subestructura de bloque> delimitado por el <símbolo de casilla>. El <identificador de bloque> identifica un bloque externo como punto extremo de canal para el <diagrama de subestructura de canal> delimitado por el <símbolo de casilla>.

```

<símbolo de canal> ::=
    <símbolo de canal 1>
    | <símbolo de canal 2>
    | <símbolo de canal 3>

```

```

<símbolo de canal 1> ::=

```



```

<símbolo de canal 2> ::=

```



```

<símbolo de canal 3> ::=

```



```


<área de lista de señales> ::=
    <símbolo de lista de señales> contains <lista de señales>

```

```

<símbolo de lista de señales> ::=

```



C1.4 Diagrama de bloque

```

<diagrama de bloque> ::=
    <símbolo de casilla>
    contains { <encabezamiento de bloque>
        { { <área de texto de bloque> }*
          { <diagrama de macro> }*
          [ <área de interacción de procesos> ]
          [ <área de subestructura de bloque> ] } set }
    is associated with { <identificador de canal> }*

```

El <identificador de canal> identifica un canal conectado a una ruta de señales en el <diagrama de bloque>. Se coloca fuera del <símbolo de casilla> cerca del punto extremo de la ruta de señales en el <símbolo de casilla>. Si el <diagrama de bloque> no contiene un <área de interacción de procesos>, tiene que contener un <área de subestructura de bloque>.

```

<encabezamiento de bloque> ::=
    BLOCK { <nombre de bloque> | <identificador de bloque> }

<área de texto de bloque> ::=
    <área de texto de sistema>

<área de interacción de procesos> ::=
    { <área de proceso>
      | <área de línea de crear>
      | <área de definición de ruta de señales> }+

<área de proceso> ::=
    <referencia gráfica de proceso>
    | <diagrama de proceso>

<referencia gráfica de proceso> ::=
    <símbolo de proceso> contains { <nombre de proceso> [<número de instancias>] }

<símbolo de proceso> ::=

```



```

<área de crear línea> ::=
    <símbolo de línea de crear>
    is connected to { <área de proceso> <área de proceso> }

<símbolo de crear línea> ::=
    ----->

<área de definición de ruta de señales> ::=
    <símbolo de ruta de señales>
    is associated with { <nombre de ruta de señales>
      { [ <identificador de canal> ]
        <área de lista de señales>
        [ <área de lista de señales> ] }set }
    is connected to
      <área de proceso> { <área de proceso> | <símbolo de casilla> }set

```

Cuando el <símbolo de ruta de señales> está conectado al <símbolo de casilla>, el <identificador de canal> identifica un canal al que está conectada la ruta de señales.

```

<símbolo de ruta de señales> ::=
    <símbolo de ruta de señales 1>
    | <símbolo de ruta de señales 2>

<símbolo de ruta de señales 1> ::=
    ----->

<símbolo de ruta de señales 2> ::=
    <----->

```

C1.5 Diagrama de proceso

```

<diagrama de proceso> ::=
    <símbolo de casilla>
    contains { <encabezamiento de proceso>
      { { <área de texto de proceso> }*
        { <área de procedimiento> }*
        { <diagrama de macro> }*
        { <área de gráfico de proceso> | <área de interacción de servicios> }set }
    [ is associated with { <identificador de ruta de señales> }+ ]

```

El <identificador de ruta de señales> identifica una ruta de señales externa conectada a una ruta de señales en el <diagrama de proceso>. Se coloca fuera del <símbolo de casilla> cerca del punto extremo de la ruta de señales en el <símbolo de casilla>.

<encabezamiento de proceso> ::=
 PROCESS { <nombre de proceso> | <identificador de proceso> }
 [<número de instancias> <fin>] [<parámetros formales>]

<área de texto de proceso> ::=
 <símbolo de texto> **contains** [<conjunto de señales de entrada válidas>]
 { <definición de señal>
 | <definición de lista de señales>
 | <definición de variable>
 | <definición de visión>
 | <definición de importación>
 | <definición de datos>
 | <definición de macro>
 | <definición de temporizador>
 | <definición de seleccionar> }*

<referencia gráfica de procedimiento> ::=
 <símbolo de procedimiento> **contains** <nombre de procedimiento>


<símbolo de procedimiento> ::=



<área de gráfico de proceso> ::=
 <área de arranque> { <área de estado> | <área de conector de entrada> }*

<área de arranque> ::=
 <símbolo de arranque> **is followed by** <área de transición>

<símbolo de arranque> ::=



<área de transición> ::=
 [<área de cadena de transición>] **is followed by**
 { <área de estado>
 | <área de estado siguiente>
 | <área de decisión>
 | <símbolo de parada>
 | <área de fusión>
 | <área de conector de salida>
 | <símbolo de retorno>
 | <área de opción de transición> }

<área de fusión> ::=
 <símbolo de fusión> **is connectd to** <símbolo de línea de flujo>

<símbolo de fusión> ::=
 <símbolo de línea de flujo>

<símbolo de línea de flujo> ::=



<área de cadena de transición> ::=
 { <área de tarea>
 | <área de salida>
 | <área de salida prioritaria>
 | <área de inicializar>
 | <área de reinicializar>
 | <área de exportación>
 | <área de petición de crear>
 | <área de llamada a procedimiento> }
 [is followed by <área de cadena de transición>]

<área de tarea> ::=
 <símbolo de tarea> contains <cuerpo de tarea>

<símbolo de tarea> ::=




<área de salida> ::=
 <símbolo de salida> contains <cuerpo de salida>

<símbolo de salida> ::=



<área de petición de crear> ::=
 <símbolo de petición de crear> contains <cuerpo de crear>

<símbolo de petición de crear> ::=



<área de llamada a procedimiento> ::=
 <símbolo de llamada a procedimiento> contains <cuerpo de llamada a procedimiento>

<símbolo de llamada a procedimiento> ::=



<área de estado> ::=
 <símbolo de estado> contains <lista de estados> is associated with
 { <área de asociación de entrada>
 | <área de asociación de entrada prioritaria>
 | <área de asociación de señal continua>
 | <área de asociación de conservación> }*


<símbolo de estado> ::=



<área de asociación de entrada> ::=
 <símbolo de asociación de trazo continuo> **is connected to** <área de entrada>

<área de entrada> ::=
 <símbolo de entrada> **contains** <lista de entradas>
 is followed by { [<área de condición habilitante>] <área de transición> }

<símbolo de entrada> ::=



<área de asociación de conservación> ::=
 <símbolo de conservación de trazo continuo> **is connected to** <área de conservación>

<área de conservación> ::=
 <símbolo de conservación> **contains** <lista de conservaciones>

<símbolo de conservación> ::=



<área de conector de entrada> ::=
 <símbolo de conector de entrada> **contains** <nombre de conector>
 is followed by <área de transición>


<símbolo de conector de entrada> ::=



<área de estado siguiente> ::=
 <símbolo de estado> **contains** <cuerpo de estado siguiente>

<área de decisión> ::=
 <símbolo de decisión> **contains** <pregunta>
 is followed by
 { { <parte respuesta gráfica> <parte otro caso gráfico> } set
 | { <parte respuesta gráfica> { <parte respuesta gráfica> } + [<parte otro caso gráfico>] } set }

<símbolo de decisión> ::=



<respuesta gráfica> ::=
 <respuesta> | (<respuesta>)

<parte respuesta gráfica> ::=
 <símbolo de línea de flujo> **is associated with** <respuesta gráfica>
 is followed by <área de transición>

<parte otro caso gráfico> ::=
 <símbolo de línea de flujo> **is associated with** ELSE
 is followed by <área de transición>

<área de inicializar> ::=
 <símbolo de tarea> **contains** <inicializar>

<área de reinicializar> ::=
 <símbolo de tarea> **contains** <reinicializar>

<símbolo de parada> ::=



<área de conector de salida> ::=
 <símbolo de conector de salida> **contains** <nombre de conector>

<símbolo de conector de salida> ::=
 <símbolo de conector de entrada>

C1.6 Diagrama de procedimiento

<diagrama de procedimiento> ::=
 <símbolo de casilla> **contains** { <encabezamiento de procedimiento>
 { { <área de texto de procedimiento> }*
 { <área de procedimiento> }*
 { <diagrama de macro> }*
 <área de gráfico de procedimiento> }set }

<encabezamiento de procedimiento> ::=
 PROCEDURE { <nombre de procedimiento> | <identificador de procedimiento> }
 [<parámetros formales de procedimiento>]

<área de procedimiento> ::=
 <referencia gráfica de procedimiento>
 | <diagrama de procedimiento>

<área de texto de procedimiento> ::=
 <símbolo de texto> **contains**
 { <definición variable>
 | <definición de datos>
 | <definición de macro>
 | <definición de seleccionar> }*

<área de gráfico de procedimiento> ::=
 <área de arranque de procedimiento> { <área de estado> | <área de conector de entrada> }*

<área de arranque de procedimiento> ::=
 <símbolo de arranque de procedimiento> **is followed by** <área de transición>

<símbolo de arranque de procedimiento> ::=



<símbolo de retorno> ::=



C1.7 *Subestructura de bloque*

<área de subestructura de bloque> ::=
 <referencia gráfica de subestructura de bloque>
 | <diagrama de subestructura de bloque>

<referencia gráfica de subestructura de bloque> ::=
 <símbolo de subestructura de bloque> **contains** <nombre de subestructura de bloque>

<símbolo de subestructura de bloque> ::=
 <símbolo de bloque>

<diagrama de subestructura de bloque> ::=
 <símbolo de casilla>
 contains { <encabezamiento de subestructura de bloque>
 { { <área de texto de subestructura de bloque> }*
 { <diagrama de macro> }*
 <área de interacción de bloque> } **set** }
 is associated with { <identificador de canal> }*

El <identificador de canal> identifica un canal conectado a un subcanal en el <diagrama de subestructura de bloque>. Se coloca fuera del <símbolo de casilla> cerca del punto extremo del subcanal en el <símbolo de casilla>.

<encabezamiento de subestructura de bloque> ::=
 SUBSTRUCTURE { <nombre de subestructura de bloque> | <identificador de subestructura de bloque> }

<área de texto de subestructura de bloque> ::=
 <área de texto de sistema>

C1.8 *Subestructura de canal*

<área de asociación de subestructura de canal> ::=
 <símbolo de asociación de trazo discontinuo>
 is connected to <área de subestructura de canal>

<área de subestructura de canal> ::=
 <referencia gráfica de subestructura de canal>
 | <diagrama de subestructura de canal>

<referencia gráfica de subestructura de canal> ::=
 <símbolo de subestructura de canal> **contains** <nombre de subestructura de canal>

<símbolo de subestructura de canal> ::=
 <símbolo de bloque>

<diagrama de subestructura de canal> ::=
 <símbolo de casilla>
 contains { <encabezamiento de subestructura de canal>
 { { <área de texto de subestructura de canal> }*
 { <diagrama de macro> }*
 <área de interacción de bloque> } **set** }
 is associated with { <identificador de bloque> | ENV }+

El <identificador de bloque> o ENV identifica un punto extremo del canal particionado. El <identificador de bloque> se coloca fuera del <símbolo de casilla> cerca del punto extremo del subcanal asociado en el <símbolo de casilla>.

<encabezamiento de subestructura de canal> ::=
 SUBSTRUCTURE { <nombre de subestructura de canal>
 | <identificador de subestructura de canal> }

<área de texto de subestructura de canal> ::=
 <área de texto de sistema>

C1.9 Macro

C1.9.1 Diagrama de macro

<diagrama de macro> ::=
 <símbolo de casilla> **contains** <encabezamiento de macro> <área de cuerpo de macro>

<encabezamiento de macro> ::=
 MACRODEFINITION <nombre de macro> [<parámetros formales de macro>]

<área de cuerpo de macro> ::=
 {{ <cualquier área> }*
 <cualquier área> [**is connected to** <cuerpo de macro puerto1>] }set
 | { <cualquier área> **is connected to** <cuerpo de macro puerto2>
 <cualquier área> **is connected to** <cuerpo de macro puerto2>
 { <cualquier área> [**is connected to** <cuerpo de macro puerto2>] }* }set

<símbolo de acceso de entrada de macro> ::=



<símbolo de acceso de salida de macro> ::=



<cuerpo de macro puerto1> ::=
 <símbolo de acceso de salida> [**is associated with** <etiqueta de macro>]
 is connected to { <símbolo de casilla>
 | <símbolo de acceso de entrada de macro>
 | <símbolo de acceso de salida de macro> }

<cuerpo de macro puerto2> ::=
 <símbolo de acceso de salida> **is associated with** <etiqueta de macro>
 is connected to { <símbolo de casilla>
 | <símbolo de acceso de entrada de macro>
 | <símbolo de acceso de salida de macro> }

<etiqueta de macro> ::=
 <nombre>

<símbolo de acceso de salida> ::=
 <símbolo de acceso de salida ficticio>
 | <símbolo de línea de flujo>
 | <símbolo de canal>
 | <símbolo de ruta de señales>
 | <símbolo de asociación de trazo continuo>
 | <símbolo de asociación de trazo discontinuo>
 | <símbolo de crear línea>

<símbolo de acceso de salida ficticio> ::=
 <símbolo de asociación de trazo continuo>

<cualquier área> ::=

- <área de texto de sistema>
- <área de interacción de bloques>
- <área de lista de señales>
- <área de bloque>
- <área de texto de bloque>
- <área de interacción de procesos>
- <referencia gráfica de procedimiento>
- <área de procedimiento>
- <área de texto de proceso>
- <área de gráfico de proceso>
- <área de fusión>
- <área de cadena de transición>
- <área de estado>
- <área de entrada>
- <área de conservación>
- <área de ampliación de texto>
- <área de asociación de subestructura de canal>
- <área de subestructura de canal>
- <área de subestructura de bloque>
- <área de entrada prioritaria>
- <área de señal continua>
- <área de conector de entrada>
- <área de estado siguiente>
- <área de proceso>
- <área de definición de canal>
- <área de línea de crear>
- <área de definición de ruta de señales>
- <referencia de gráfica de proceso>
- <diagrama de proceso>
- <área de arranque>
- <área de salida>
- <área de inicializar>
- <área de reinicializar>
- <área de exportación>
- <área de salida prioritaria>
- <área de tarea>
- <área de petición de crear>
- <área de llamada de procedimiento>
- <área de decisión>
- <área de conector de salida>
- <área de texto de procedimiento>
- <área de gráfico de procedimiento>
- <área de arranque de procedimiento>
- <área de texto de subestructura de bloque>
- <área de interacción de bloques>
- <área de servicio>
- <área de definición de ruta de señales de servicio>
- <área de texto de servicio>
- <área de gráfico de servicio>
- <área de arranque de servicio>
- <área de comentario>
- <área de llamada a macro>

C1.9.2 Llamada a macro


<área de llamada a macro> ::=

```

< símbolo de llamada a macro > contains { < nombre de macro > [< cuerpo de llamada a macro > ]}
[is connected to
{ < llamada a macro puerto1 > | < llamada a macro puerto2 > { < llamada a macro puerto2 > }+ } ]

```

<símbolo de llamada a macro > ::=



<llamada a macro puerto1 > ::=
 <símbolo de acceso de entrada > [is associated with <etiqueta de macro >]
 is connected to <cualquier área >

<llamada a macro puerto2 > ::=
 <símbolo de acceso de entrada > is associated with <etiqueta de macro >
 is connected to <cualquier área >

<símbolo de acceso de entrada > ::=
 <símbolo de acceso de entrada ficticio >
 | <símbolo de línea de flujo >
 | <símbolo de canal >
 | <símbolo de ruta de señales >
 | <símbolo de asociación de trazo continuo >
 | <símbolo de asociación de trazo discontinuo >
 | <símbolo de línea de crear >

<símbolo de acceso de entrada ficticio > ::=
 <símbolo de asociación de trazo continuo >

C1.10 *Sistemas genéricos*

C1.10.1 *Definición opcional*


<área de opción > ::=
 <símbolo de opción > contains
 { SELECT IF (expresión simple booleana)
 { <área de bloque >
 | <área de subestructura de canal >
 | <área de texto de sistema >
 | <área de texto de bloque >
 | <área de texto de proceso >
 | <área de texto de procedimiento >
 | <área de texto de subestructura de bloque >
 | <área de texto de subestructura de canal >
 | <área de texto de servicio >
 | <diagrama de macro >
 | <área de opción >
 | <área de proceso >
 | <área de definición de ruta de señales >
 | <área de línea de crear >
 | <área de procedimiento >
 | <área de servicio >
 | <área de definición de ruta de señales de servicio > }+ } }

El <símbolo de opción > es un polígono de trazo discontinuo con esquinas de trazo continuo, por ejemplo:



C1.10.1 *Transición opcional*

<área de opción de transición> ::=
 <símbolo de opción de transición> **contains** { <pregunta alternativa> }
 is followed by { <opción acceso de salida1> { <opción acceso de salida1> | <opción acceso de salida2> }
 { <opción acceso de salida1> }* }set
<símbolo de opción de transición> ::=



<opción acceso de salida1> ::=
 <símbolo de línea de flujo> **is associated with** <respuesta gráfica>
 is followed by { <área de transición> | <área de fusión> }
<opción acceso de salida2> ::=
 <símbolo de línea de flujo> **is associated with** ELSE
 is followed by { <área de transición> | <área de fusión> }

C1.11 *Servicio*

C1.11.1 *Descomposición de servicio*

<área de interacción de servicios> ::=
 { <área de servicio> | <área de definición de ruta de señales de servicio> }+
<área de servicio> ::=
 <referencia gráfica de servicio>
 | <diagrama de servicio>
<referencia gráfica de servicio> ::=
 <símbolo de servicio> **contains** <nombre de servicio>
<símbolo de servicio> ::=



<área de definición de ruta de señales de servicio> ::=
 <símbolo de ruta de señales>
 is associated with { <nombre de ruta de señales de servicio>
 [<identificador de ruta de señales>]
 <área de lista de señales> [<área de lista de señales>] }set
 is connected to { <área de servicio>
 { <área de servicio> | <símbolo de casilla> } }set

Cuando el <símbolo de ruta de señales> está conectado al <símbolo de casilla>, el identificador <de ruta de señales> identifica una ruta de señales externa a la cual se conecta la ruta de señales.

C1.11.2 *Diagrama de servicio*

<diagrama de servicio> ::=
 <símbolo de casilla> **contains**
 { <encabezamiento de servicio>
 { { <área de texto de servicio> }*
 { <área de procedimiento> }*
 { <diagrama macro> }*
 <área de gráfico de servicio> }set }
<encabezamiento de servicio> ::=
 SERVICE { <nombre de servicio> | <identificador de servicio> }

<área de texto de servicio> ::=
 <símbolo de texto> **contains**
 { <definición variable>
 | <definición de visión>
 | <definición de importación>
 | <definición de datos>
 | <definición de macro>
 | <definición de temporizador>
 | <definición de seleccionar> }*

<área de gráfico de servicio> ::=
 <área de gráfico de proceso>

<área de asociación de entrada prioritaria> ::=
 <símbolo de asociación de trazo continuo> **is connected to** <área de entrada prioritaria>

<área de entrada prioritaria> ::=
 <símbolo de entrada prioritaria> **contains** <lista de entradas prioritarias>

<símbolo de entrada prioritaria> ::=



<área de salida prioritaria> ::=
 <símbolo de salida prioritaria> **contains** <cuerpo de salida prioritaria>

<símbolo de salida prioritaria> ::=



C1.12 Señal continua

<área de asociación de señal continua> ::=
 <símbolo de asociación de trazo continuo> **is connected to** <área de señal continua>

<área de señal continua> ::=
 <símbolo de condición habilitante>
contains{ <expresión booleana> [<fin> PRIORITY<nombre de literal entero>]}
is followed by<área de transición>

C1.13 Condición habilitante

<área de condición habilitante> ::=
 <área de condición habilitante> **contains** <expresión booleana>

<símbolo de condición habilitante> ::=



C1.14 Exportación

<área de exportación> ::=
 <símbolo de tarea> **contains**<exportación>

SERIES DE RECOMENDACIONES DEL UIT-T

Serie A	Organización del trabajo del UIT-T
Serie B	Medios de expresión: definiciones, símbolos, clasificación
Serie C	Estadísticas generales de telecomunicaciones
Serie D	Principios generales de tarificación
Serie E	Explotación general de la red, servicio telefónico, explotación del servicio y factores humanos
Serie F	Servicios de telecomunicación no telefónicos
Serie G	Sistemas y medios de transmisión, sistemas y redes digitales
Serie H	Sistemas audiovisuales y multimedia
Serie I	Red digital de servicios integrados
Serie J	Transmisiones de señales radiofónicas, de televisión y de otras señales multimedia
Serie K	Protección contra las interferencias
Serie L	Construcción, instalación y protección de los cables y otros elementos de planta exterior
Serie M	RGT y mantenimiento de redes: sistemas de transmisión, circuitos telefónicos, telegrafía, facsimil y circuitos arrendados internacionales
Serie N	Mantenimiento: circuitos internacionales para transmisiones radiofónicas y de televisión
Serie O	Especificaciones de los aparatos de medida
Serie P	Calidad de transmisión telefónica, instalaciones telefónicas y redes locales
Serie Q	Conmutación y señalización
Serie R	Transmisión telegráfica
Serie S	Equipos terminales para servicios de telegrafía
Serie T	Terminales para servicios de telemática
Serie U	Conmutación telegráfica
Serie V	Comunicación de datos por la red telefónica
Serie X	Redes de datos y comunicación entre sistemas abiertos
Serie Y	Infraestructura mundial de la información y aspectos del protocolo Internet
Serie Z	Lenguajes y aspectos generales de soporte lógico para sistemas de telecomunicación